1. Elements of linear algebra

Contents

1.1. Solving systems of linear equations

Consider the following system of linear equations:

$$\begin{cases} a_{11}\,x_1 + a_{12}\,x_2 + \ldots + a_{1n}\,x_n &= b_1 \\ a_{21}\,x_1 + a_{22}\,x_2 + \ldots + a_{2n}\,x_n &= b_2 \\ \quad \ldots & \quad \ldots \\ a_{m1}\,x_1 + a_{m2}\,x_2 + \ldots + a_{mn}\,x_n &= b_m\,. \end{cases} \tag{1.1}$$

Here $a_{ij} \in \mathbb{R}$, $b_j \in \mathbb{R}$.

The system (1.1) can be written in the form

$$A \cdot X = B \tag{1.2}$$

where

$$A = \begin{pmatrix} a_{11} & a_{12} & \ldots & a_{1n} \\ a_{21} & a_{22} & \ldots & a_{2n} \\ \ldots & \ldots & \ldots & \ldots \\ a_{m1} & a_{m2} & \ldots & a_{mn} \end{pmatrix}, \quad X = \begin{pmatrix} x_1 \\ x_2 \\ \ldots \\ x_n \end{pmatrix}, \quad B = \begin{pmatrix} b_1 \\ b_2 \\ \ldots \\ b_m \end{pmatrix}.$$

It is well known that if the system (1.1) or (1.2) is consistent, that is, there is a vector (a particular solution)

$$X^{(0)} = \begin{pmatrix} x_1^{(0)} \\ x_2^{(0)} \\ \ldots \\ x_n^{(0)} \end{pmatrix},$$

satisfying (1.2), then each solution X of (1.2) can be written in the form

$$X = X^{(0)} + t_1 X^{(1)} + t_2 X^{(2)} + \ldots + t_r X^{(r)}\,. \tag{1.3}$$

Here $t_1 \in \mathbb{R}$, …, $t_r \in \mathbb{R}$ are free variables and $\{X^{(1)}, X^{(2)}, \ldots, X^{(r)}\}$ is a fundamental system of solutions for the homogeneous system

$$A \cdot X = 0\,, \tag{1.4}$$

$$X^{(j)} = \begin{pmatrix} x_1^{(j)} \\ x_2^{(j)} \\ \ldots \\ x_n^{(j)} \end{pmatrix}, \qquad j = 1, \ldots, r\,.$$

Thus, the general solution of (1.1) can be represented as follows

$$\begin{aligned} x_1 &= x_1^{(0)} + t_1 x_1^{(1)} + t_2 x_1^{(2)} + \ldots + t_r x_1^{(r)} \\ x_2 &= x_2^{(0)} + t_1 x_2^{(1)} + t_2 x_2^{(2)} + \ldots + t_r x_2^{(r)} \\ \ldots \\ x_n &= x_n^{(0)} + t_1 x_n^{(1)} + t_2 x_n^{(2)} + \ldots + t_r x_n^{(r)} \ . \end{aligned} \tag{1.5}$$

The solution (1.5) can be found, for instance, due to Gauss-Jordan algorithm.

Example 1.1.

$$\left\{ \begin{array}{rrrcl} x_1 & +2\,x_2 & +3\,x_3 & = & 10 \\ 4x_1 & +5\,x_2 & +6\,x_3 & = & 11 \\ 7x_1 & +8\,x_2 & +9\,x_3 & = & 12\ . \end{array} \right. \tag{1.6}$$

The system (1.6) can be expressed in the matrix form as follows

$$\left(\begin{array}{ccc|c} 1 & 2 & 3 & 10 \\ 4 & 5 & 6 & 11 \\ 7 & 8 & 9 & 12 \end{array} \right) \ . \tag{1.7}$$

Using the Gauss-Jordan algorithm one can reduce (1.7) first to the *row echelon* form

$$\left(\begin{array}{rrr|r} 1 & 2 & 3 & 10 \\ 0 & -3 & -6 & -29 \end{array} \right) \tag{1.8}$$

and then to the *reduced row echelon* form

$$\left(\begin{array}{rrr|r} 1 & 0 & -1 & -28/3 \\ 0 & 1 & 2 & 29/3 \end{array} \right) . \tag{1.9}$$

The variable x_3 can be regarded as a free variable, $x_3 = t_1$. It follows from (1.9) that

$$\begin{aligned} x_1 &= -\frac{28}{3} + 1 \cdot t_1 \\ x_2 &= \frac{29}{3} - 2 \cdot t_1 \\ x_3 &= 1 \cdot t_1 \end{aligned}$$

is the solution of (1.6).

Computations in MAPLE

```
> restart; with(LinearAlgebra):
> A := Matrix([[1,2,3],[4,5,6],[7,8,9]]);
```

$$A := \left[\begin{array}{ccc} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{array} \right]$$

```
> GaussianElimination(A);
```

$$\left[\begin{array}{rrr} 1 & 2 & 3 \\ 0 & -3 & -6 \\ 0 & 0 & 0 \end{array} \right]$$

```
> ReducedRowEchelonForm(<A>);
```

$$\begin{bmatrix} 1 & 0 & -1 \\ 0 & 1 & 2 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> B:=Vector([10,11,12]);
```

$$B := \begin{bmatrix} 10 \\ 11 \\ 12 \end{bmatrix}$$

```
> X:=LinearSolve(A, B,free='t'); # the general solution
```

$$X := \begin{bmatrix} -\frac{28}{3} + t_3 \\ \frac{29}{3} - 2\,t_3 \\ t_3 \end{bmatrix}$$

Example 1.2.

$$\begin{cases} x_1 & +2\,x_2 & +3\,x_3 & = & 10 \\ 4x_1 & +5\,x_2 & +6\,x_3 & = & 11 \\ 7x_1 & +8\,x_2 & +9\,x_3 & = & 13\ . \end{cases} \tag{1.10}$$

If we try to solve this system with Maple we obtain the following result

Computations in MAPLE

```
> with(LinearAlgebra):
> A := Matrix([[1,2,3],[4,5,6],[7,8,9]]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

```
> B:=Vector([10,11,13]);
```

$$B := \begin{bmatrix} 10 \\ 11 \\ 13 \end{bmatrix}$$

```
> X:=LinearSolve(A, B,free='t');

Error, (in LinearAlgebra:-LA_Main:-LinearSolve) inconsistent
system
```

Thus, the system (1.10) is inconsistent.

1.2. Diagonal form of a square matrix

Consider a matrix $A = (a_{ij})_{n\times n}$ which is a square $n \times n$ matrix with real (or complex) entries. The main objective is to reduce this matrix to a diagonal form. It means that one should find a square non-degenerate $n \times n$ matrix C, a *transition* matrix, and a matrix D for which the equality

$$C^{-1} \cdot A \cdot C = D \tag{1.11}$$

holds. Here D is a diagonal matrix,

$$D = \begin{pmatrix} \lambda_1 & 0 & 0 & \dots & 0 \\ 0 & \lambda_2 & 0 & \dots & 0 \\ \vdots & & \ddots & & \vdots \\ & & & & \\ 0 & 0 & 0 & \dots & \lambda_n \end{pmatrix} .$$

Remark. Note that the problem of diagonalization can have no solution even if we consider the complex matrices. For instance, the matrix

$$M = \begin{pmatrix} 0 & 1 \\ 0 & 0 \end{pmatrix}$$

is not diagonalizable. In general, there is a matrix equality of type (1.11) with a matrix D having *Jordan normal form*. The diagonal form is a particular case of Jordan normal form.

Eigenvectors and eigenvalues of a square matrix

Let us suppose that matrices C and D exist for a given matrix A. In order to obtain the diagonal matrix D we should find all *eigenvalues* of A. Then matrix C can be determined in terms of *eigenvectors* of A (see the general case below).

Definition. A *non-trivial* vector

$$X = \begin{pmatrix} x_1 \\ x_2 \\ \vdots \\ x_n \end{pmatrix} , \quad X \neq 0 ,$$

is called *eigenvector* of a square $n \times n$ matrix A if there is a number $\lambda \in \mathbb{C}$ such that the equality

$$A \cdot X = \lambda X \tag{1.12}$$

holds. In this case λ is called *eigenvalue* of A.

Note that the equality (1.12) can be rewritten in the form

$$(A - \lambda E) \cdot X = 0 \tag{1.13}$$

where E is the identity matrix.

The matrix equality (1.13) can be regarded as a homogeneous system of linear equations. It is well known that it has a non-trivial solution if and only if

$$\det(A - \lambda E) = 0 .$$

Recall that the polynomial $\chi_A(\lambda) = \det(A - \lambda E)$ is called *characteristic* polynomial of a matrix A.

Thus, a complex number λ is an eigenvalue of a matrix A if and only if λ is a root of *characteristic* polynomial of A.

Example 1.3. Find the eigenvalues of the matrix

$$A = \begin{pmatrix} -5 & -8 & -16 \\ -18 & -25 & -46 \\ 12 & 17 & 32 \end{pmatrix} .$$

The characteristic polynomial

$$\det\begin{pmatrix} -5-\lambda & -8 & -16 \\ -18 & -25-\lambda & -46 \\ 12 & 17 & 32-\lambda \end{pmatrix} = -6+5\lambda+2\lambda^2-\lambda^3 \, .$$

The roots of the characteristic polynomial, the eigenvalues of A, are equal to -2, 1, 3 .

Remark. The diagonal entries of the matrix D in (1.11) are eigenvalues of A. In view of (1.11) there exists a non-degenerate matrix C such that

$$C^{-1}\cdot\begin{pmatrix} -5 & -8 & -16 \\ -18 & -25 & -46 \\ 12 & 17 & 32 \end{pmatrix}\cdot C = \begin{pmatrix} -2 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 3 \end{pmatrix} \, .$$

It can be showed that as rows of such a transition matrix C the corresponding eigenvectors can be chosen (see the general case below).

Computations in MAPLE

```
> restart; with(LinearAlgebra):
> A := Matrix([[-5,-8,-16],[-18,-25,-46],[12,17,32]]);
```

$$A := \begin{bmatrix} -5 & -8 & -16 \\ -18 & -25 & -46 \\ 12 & 17 & 32 \end{bmatrix}$$

```
> CharacteristicPolynomial(A,lambda);
```

$$6-5\,\lambda-2\,\lambda^2+\lambda^3$$

```
> solve(%,lambda);
```

$$1,\ -2,\ 3$$

```
> L:=Eigenvalues(A,output='list');
```

$$L := [1,\ -2,\ 3]$$

```
> E:=Matrix(3,3,shape=identity);   #The identity matrix
```

$$E := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> B:=Vector(3,shape=zero);
```

$$B := \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

```
> X1:=LinearSolve(A-L[2]*E,B,free='t');
> #this is eigenvector for the
> eigenvalue L[2]=-2
```

$$X1 := \begin{bmatrix} 0 \\ -2\,t_3 \\ t_3 \end{bmatrix}$$

```
> X2:=LinearSolve(A-L[1]*E,B,free='t');
> #this is eigenvector for the
> eigenvalue L[1]=1
```

$$X2 := \begin{bmatrix} -4\,t_3 \\ t_3 \\ t_3 \end{bmatrix}$$

```
> X3:=LinearSolve(A-L[3]*E,B,free='t');
> #this is eigenvector for the
> eigenvalue L[3]=3
```

$$X3 := \begin{bmatrix} -t_3 \\ -t_3 \\ t_3 \end{bmatrix}$$

```
> t[3]:=1;
```

$$t_3 := 1$$

```
> C:=Matrix([X1,X2,X3]); # A transition matrix
```

$$C := \begin{bmatrix} 0 & -4 & -1 \\ -2 & 1 & -1 \\ 1 & 1 & 1 \end{bmatrix}$$

```
> Diag:=C^(-1).A.C; #  this is to check the result
```

$$Diag := \begin{bmatrix} -2 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 3 \end{bmatrix}$$

Thus, we have obtained a transition matrix

$$C = \begin{pmatrix} 0 & -4 & -1 \\ -2 & 1 & -1 \\ 1 & 1 & 1 \end{pmatrix}.$$

Remark. The characteristic polynomial is defined up to sign ± 1.

The diagonal form and the transition matrix in general case

Suppose that A is a diagonalizable matrix and let

$$\chi_A(\lambda) = \det(A - \lambda E) = \pm(\lambda - \lambda_1)^{k_1} \cdot \ldots \cdot (\lambda - \lambda_s)^{k_s} \tag{1.14}$$

be a decomposition of characteristic polynomial of A over $\mathbb{C}$. The eigenvalues $\lambda_1, \ldots, \lambda_s$ are pairwise distinct and $k_1 + \ldots + k_s = n$ where n is the dimension of A.

The following theorem provides a necesary and sufficient condition for the existence of diagonal form of A.

Theorem. In previous notation suppose that for each $l = 1, \ldots, s$ the system

$$(A - \lambda_l E)X = 0$$

has k_l linearly independent solutions $X_{l,1}, \ldots, X_{l,k_l}$ which are eigenvectors of A corresponding to the eigenvalue λ_l.

Then $C^{-1} \cdot A \cdot C = D$ where

$$D = \begin{pmatrix} \lambda_1 & & & & & & \\ & \ddots & & & 0 & & \\ & & \lambda_1 & & & & \\ & & & \ddots & & & \\ & & & & \lambda_s & & \\ & & 0 & & & \ddots & \\ & & & & & & \lambda_s \end{pmatrix}$$

is a diagonal matrix with eigenvalues of A as diagonal entries (λ_l appears k_l times) and the columns of C are linearly independent eigenvectors $X_{1,1}, \ldots, X_{1,k_1}, \ldots, X_{s,1}, \ldots, X_{s,k_s}$.

In particular, if the characteristic polynomial of A has no multiple roots (i.e. all $k_l = 1$ in (1.14), the total number of different eigenvalues is equal to n) then the diagonal form of A exists.

Example 1.4. Let us find the transition matrix C in the equality

$$C^{-1} \cdot \begin{pmatrix} -5 & -8 & -16 \\ -18 & -25 & -46 \\ 12 & 17 & 32 \end{pmatrix} \cdot C = \begin{pmatrix} -2 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 3 \end{pmatrix} .$$

of Example 1.3.

For the eigenvalue $\lambda_1 = -2$ we obtain an eigenvector $X_1 = \begin{pmatrix} 0 \\ -2 \\ 1 \end{pmatrix}$, for the eigenvalue $\lambda_2 = 1$ – an eigenvector $X_2 = \begin{pmatrix} -4 \\ 1 \\ 1 \end{pmatrix}$ and finally, for the eigenvalue $\lambda_3 = 3$ – an eigenvector $X_3 = \begin{pmatrix} -1 \\ -1 \\ 1 \end{pmatrix}$. Then

$$C = \begin{pmatrix} 0 & -4 & -1 \\ -2 & 1 & -1 \\ 1 & 1 & 1 \end{pmatrix}$$

Computations in MAPLE

```
> restart; with(LinearAlgebra):
> A := «-5,-18,12>|<-8,-25,17>|<-16,-46,32»;
```

$$A := \begin{bmatrix} -5 & -8 & -16 \\ -18 & -25 & -46 \\ 12 & 17 & 32 \end{bmatrix}$$

```
> K:=CharacteristicPolynomial(A,lambda);
```

$$K := 6 - 5\,\lambda - 2\,\lambda^2 + \lambda^3$$

```
> factor(K);
```

$$(\lambda - 1)\,(\lambda - 3)\,(\lambda + 2)$$

```
> (lambda,C):=Eigenvectors(A);
```

$$\lambda, C := \begin{bmatrix} 3 \\ -2 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 & 0 & -4 \\ -1 & -2 & 1 \\ 1 & 1 & 1 \end{bmatrix}$$

```
> C;
```

$$\begin{bmatrix} -1 & 0 & -4 \\ -1 & -2 & 1 \\ 1 & 1 & 1 \end{bmatrix}$$

```
> C^(-1); #this is the inverse matrix
```

$$\begin{bmatrix} 3 & 4 & 8 \\ -2 & -3 & -5 \\ -1 & -1 & -2 \end{bmatrix}$$

```
> Diag:=C^(-1).A.C;
```

$$Diag := \begin{bmatrix} 3 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

1.3. The Jordan normal form of a square matrix

Sometimes the characteristic polynomial of a square matrix A has multiple roots and A has no diagonal form. However, instead of diagonal form there is a Jordan normal form of A, that is, there is a matrix equality

$$C^{-1} \cdot A \cdot C = J$$

with a block diagonal matrix J and some non-degenerate matrix C. The matrix J is of block diagonal type

$$\begin{pmatrix} J_{m_1}(\lambda_1) & & & & & 0 & \\ & \ddots & & & & & \\ & & J_{m_t}(\lambda_1) & & & & \\ & & & \ddots & & & \\ 0 & & & & J_{m_j}(\lambda_s) & & \\ & & & & & \ddots & \\ & & & & & & J_{m_k}(\lambda_s) \end{pmatrix} \tag{1.15}$$

where

$$J_p(\lambda) = \begin{pmatrix} \lambda & 1 & 0 & 0 & \dots & 0 \\ 0 & \lambda & 1 & 0 & \dots & 0 \\ 0 & 0 & \lambda & 1 & \dots & 0 \\ 0 & 0 & 0 & \lambda & \dots & 0 \\ \vdots & & & & \ddots & 1 \\ 0 & 0 & & & 0 & \lambda \end{pmatrix}$$

is a $p \times p$ matrix with entries equal to λ on the diagonal and entries equal to 1 just immediately over the diagonal. The other entries are trivial. $J_p(\lambda)$ is called a Jordan cell.

In the form (1.15) the number and the dimensions of Jordan cells for each eigenvalue λ_k are uniquely defined by the matrix A.

Remark. The diagonal form is a particular case of Jordan form. In the diagonal form all Jordan cells are 1×1 matrices.

Example 1.5. Let us find the Jordan form of a matrix

$$A = \begin{pmatrix} 23 & 28 & 52 \\ 10 & 16 & 27 \\ -14 & -19 & -34 \end{pmatrix} .$$

The characteristic polynomial is equal (up to the sign) to

$$(\lambda + 1)(\lambda - 3)^2 .$$

The Jordan form

$$J = \begin{pmatrix} -1 & 0 & 0 \\ 0 & 3 & 1 \\ 0 & 0 & 3 \end{pmatrix}$$

has one 1-cell $J_1(-1)$ and one 2-cell $J_2(3)$. Note that A has no diagonal form.

Computations in MAPLE

```
> with(LinearAlgebra):
> A:=Matrix([[23,28,52],[10,16,27],[-14,-19,-34]]);
```

$$A := \begin{bmatrix} 23 & 28 & 52 \\ 10 & 16 & 27 \\ -14 & -19 & -34 \end{bmatrix}$$

```
> CharacteristicPolynomial(A,lambda);
```

$$9 + 3\,\lambda - 5\,\lambda^2 + \lambda^3$$

```
> factor(%);
```

$$(\lambda + 1)\,(\lambda - 3)^2$$

```
> Eigenvalues(A);
```

$$\begin{bmatrix} -1 \\ 3 \\ 3 \end{bmatrix}$$

```
> J:=JordanForm(A);
```

$$J := \begin{bmatrix} -1 & 0 & 0 \\ 0 & 3 & 1 \\ 0 & 0 & 3 \end{bmatrix}$$

1.4. The Gram-Schmidt orthogonalization process

Consider the euclidian space $\mathbb{R}^n$, that is, the vector space $\mathbb{R}^n$ equipped with the inner product. The inner product two vectors

$$X = (x_1, \ldots, x_n)\ ,\ Y = (y_1, \ldots, y_n)$$

is defined as follows

$$X \cdot Y = x_1\, y_1 + \ldots + x_n\, y_n \ .$$

Recall that the euclidian norm (the length) $\|X\|$ of a vector X is the following number

$$\|X\| = \sqrt{X \cdot X} = \sqrt{x_1^2 + \ldots + x_n^2}\ .$$

Definition. A vector $X \in \mathbb{R}^n$ is called *normalized* if $\|X\| = 1$.

Definition. Two vectors $X \in \mathbb{R}^n$ and $Y \in \mathbb{R}^n$ are called *orthogonal* if $X \cdot Y = 0$. A system of vectors S is called *orthogonal* if all vectors of the system are pairwise orthogonal. A system of vectors S is called *orthonormal* if it is orthogonal and each vector in S is normalized.

Note that any orthogonal system consists of linearly independent vectors.

In linear algebra the following orthogonalization problem is often needed to be solved.

Orthogonalization problem. For a given system of vectors

$$S = \{V_1, \ldots, V_k\}\,, \quad V_i \in \mathbb{R}^n\,,$$

to find an orthogonal (orthonormal) system

$$\Sigma = \{U_1, \ldots, U_m\}\,, \quad m \leqslant k\ ,$$

such that the linear subspaces spanned by S and Σ coincide:

$$\langle V_1, \ldots, V_k \rangle = \langle U_1, \ldots, U_m \rangle\ .$$

This problem can be solved with the help of the Gram-Schmidt orthogonalization process.

The Gram-Schmidt orthogonalization process. This process works inductively. Without loss of generality we may assume that there is no trivial vector in S.

Step 1. Let $U_1 = V_1$.

Step 2. It follows from the first step that $\langle V_1 \rangle = \langle U_1 \rangle$. We will seek a vector U_2 in the form

$$U_2 = V_2 - a_{21} U_1 \tag{1.16}$$

The condition $U_2 \cdot U_1 = 0$ is required and consequently,

$$0 = U_2 \cdot U_1 = (V_2 - a_{21}U_1) \cdot U_1 = V_2 \cdot U_1 - a_{21}\, U_1 \cdot U_1\ .$$

It follows that

$$a_{21} = \frac{V_2 \cdot U_1}{U_1 \cdot U_1}\ ,$$

U_1 and U_2 are orthogonal.

If $U_2 \neq 0$ then it is included in the system Σ (otherwise we omit trivial vector). Now we go on to the next step.

Step 3. Due to (1.16) $\langle V_1, V_2 \rangle = \langle U_1, U_2 \rangle$. We will seek a vector U_3 in the form

$$U_3 = V_3 - a_{31} U_1 - a_{32} U_2 \tag{1.17}$$

The conditions $U_3 \cdot U_1 = 0$ and $U_3 \cdot U_2 = 0$ are required and consequently,

$$0 = U_3 \cdot U_1 = (V_3 - a_{31}U_1 - a_{32}U_2) \cdot U_1 = V_3 \cdot U_1 - a_{31}\, U_1 \cdot U_1 \ ,$$

$$0 = U_3 \cdot U_2 = (V_3 - a_{31}U_1 - a_{32}U_2) \cdot U_1 = V_3 \cdot U_2 - a_{32}\, U_2 \cdot U_2$$

since $U_1 \cdot U_2 = U_2 \cdot U_1 = 0$. It follows that

$$a_{31} = \frac{V_3 \cdot U_1}{U_1 \cdot U_1} \ , \quad a_{32} = \frac{V_3 \cdot U_2}{U_2 \cdot U_2}$$

and U_1 , U_2, U_3 are pairwise orthogonal. If $U_3 \neq 0$ then it is included in the system Σ and the process is going on.

. . .

Step l. Suppose that we have already determined pairwise orthogonal vectors $U_1, U_2, \ldots, U_{s-1}$ such that

$$\langle V_1, V_2, \ldots V_{l-1} \rangle = \langle U_1, U_2, \ldots, U_{s-1} \rangle \ .$$

Then as above

$$U_s = V_l - a_{l,1}U_1 - \ldots - a_{l,s-1}U_{s-1} \tag{1.18}$$

where

$$a_{l,j} = \frac{V_l \cdot U_j}{U_j \cdot U_j} \ . \tag{1.19}$$

If $U_s \neq 0$ then it is included in Σ and so on.

Since the number of vectors is finite then this process will terminate after k steps and we obtain a desired orthogonal system Σ. The following transform (normalization) is to be done in order to make Σ orthonormal:

$$\{U_1, \ldots, U_m\} \to \left\{ \frac{U_1}{\| U_1 \|}, \ldots, \frac{U_m}{\| U_m \|} \right\} \ .$$

The Gram-Schmidt orthogonalization process from a geometrical point of view.

The l-th step of the Gram-Schmidt orthogonalization process can be viewed as follows (see Figure 1 below). Denote $L_{s-1} = \langle U_1, U_2, \ldots, U_{s-1} \rangle$. A given vector V_l should be represented as the (unique) sum

$$V_l = U_s + (V_l - U_s)$$

where U_s is orthogonal to the subspace L_{s-1} (the vector U_s is called *the orthogonal component* of V_l with respect to L_{s-1}) and $V_l - U_s \in L_{s-1}$ (the vector $V_l - U_s$ is the orthogonal projection of V_l onto L_{s-1}). It is sufficient to determine

$$V_l - U_s = a_{l,1}U_1 + \ldots + a_{l,s-1}U_{s-1} \tag{1.20}$$

(cf. (1.18)) since

$$U_s = V_l - (V_l - U_s) \, .$$

The coefficients $a_{l,j}$ in (1.20) are defined by the formula (1.19).



Fig. 1

Note that the distance between the endpoint of V_l and the subspace L_{s-1} is equal to the length $\| U_s \|$.

Example 1.6. Determine the orthogonal basis of the subspace $L \subset \mathbb{R}^4$ spanned by the vectors

$$V_1 = (1, 2, 2, -1)\,, \quad V_2 = (1, 1, -5, 3)\,, \quad V_3 = (4, 5, -13, 8)\,.$$

Solution. Apply the Gram-Schmidt orthogonalization process to the system $S = \{V_1, V_2, V_3\}$.

Step 1.

$$U_1 = V_1 = (1, 2, 2, -1)\ .$$

Step 2.

$$U_2 = V_2 - a_{21} U_1$$

where

$$a_{21} = \frac{V_2 \cdot U_1}{U_1 \cdot U_1} = \frac{1 \cdot 1 + 2 \cdot 1 + 2 \cdot (-5) + (-1) \cdot 3}{1^2 + 2^2 + 2^2 + (-1)^2} = -1\ .$$

Consequently,

$$U_2 = V_2 - (-1) U_1 = (2, 3, -3, 2)\ .$$

Step 3.

$$U_3 = V_3 - a_{31} U_1 - a_{32} U_2$$

where

$$a_{31} = \frac{V_3 \cdot U_1}{U_1 \cdot U_1} = -2\ , \quad a_{32} = \frac{V_3 \cdot U_2}{U_2 \cdot U_2} = 3\ .$$

Hence,

$$U_3 = V_3 - (-2) U_1 - 3 U_2 = 0\ .$$

Thus, U_3 is not included in the orthogonal basis and the subspace L is 2-dimensional. The orthogonal basis is

$$\Sigma = \{U_1, U_2\}\ .$$

The normalized system Σ_{norm} is

$$\left\{ \frac{U_1}{\| U_1 \|}\, , \frac{U_2}{\| U_2 \|} \right\} = \left\{ \left(\frac{1}{\sqrt{10}}, \frac{2}{\sqrt{10}}, \frac{2}{\sqrt{10}}, -\frac{1}{\sqrt{10}} \right)\, , \left(\frac{2}{\sqrt{26}}, \frac{3}{\sqrt{26}}, -\frac{3}{\sqrt{26}}, \frac{2}{\sqrt{26}} \right) \right\}\ .$$

Computations in MAPLE

```
> with(LinearAlgebra):
```

```
> v1:=Vector[row]([1,2,2,-1]);
```

$$v1 := [1, 2, 2, -1]$$

```
> v2:=Vector[row]([1,1,-5,3]);
```

$$v2 := [1, 1, -5, 3]$$

```
> v3:=Vector[row]([4,5,-13,8]);
```

$$v3 := [4, 5, -13, 8]$$

```
> Basis([v1,v2,v3]); #this is to determine
> the dimension of subspace <v1,v2,v3>
```

$$[[1, 2, 2, -1], [1, 1, -5, 3]]$$

```
> res:=GramSchmidt([v1,v2,v3]);
```

$$res := [[1, 2, 2, -1], [2, 3, -3, 2]]$$

```
> A:=Matrix([[res[1]],[res[2]]]);
```

$$A := \begin{bmatrix} 1 & 2 & 2 & -1 \\ 2 & 3 & -3 & 2 \end{bmatrix}$$

```
> C:=A.Transpose(A);  # this must be the
> diagonal  matrix of inner products
```

$$C := \begin{bmatrix} 10 & 0 \\ 0 & 26 \end{bmatrix}$$

```
> Normalize(res[1],2); #  2 stands for the euclidian norm
```

$$\left[\frac{1}{10}\sqrt{10}, \frac{1}{5}\sqrt{10}, \frac{1}{5}\sqrt{10}, -\frac{1}{10}\sqrt{10}\right]$$

```
> Normalize(res[2],2);
```

$$\left[\frac{1}{13}\sqrt{26}, \frac{3}{26}\sqrt{26}, -\frac{3}{26}\sqrt{26}, \frac{1}{13}\sqrt{26}\right]$$

```
> res_norm:=GramSchmidt([v1,v2,v3],normalized);
```

$$res\_norm := \left[\left[\frac{1}{10}\sqrt{10}, \frac{1}{5}\sqrt{10}, \frac{1}{5}\sqrt{10}, -\frac{1}{10}\sqrt{10}\right], \left[\frac{1}{13}\sqrt{26}, \frac{3}{26}\sqrt{26}, -\frac{3}{26}\sqrt{26}, \frac{1}{13}\sqrt{26}\right]\right]$$

Example 1.7. Determine the distance d between the (endpoint of the) vector $X = (-1, 5, 1, -1)$ and the subspace $L \subset \mathbb{R}^4$ given as the subspace of solutions of the linear homogeneous system

$$L: \begin{cases} 4x_2 - x_3 + 3x_4 = 0 \\ 2x_1 + 2x_2 + x_3 + x_4 = 0\,. \end{cases}$$

Solution. First of all, let us find a basis of L. As a basis a fundamental system of solutions of the above linear system can be taken.

Computations in MAPLE

```
> restart;with(LinearAlgebra):
> A := Matrix([[0,4,-1,3],[2,2,1,1]]);
```

$$A := \begin{bmatrix} 0 & 4 & -1 & 3 \\ 2 & 2 & 1 & 1 \end{bmatrix}$$

```
> B:=Vector([0,0]);
```

$$B := \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

```
> X:=LinearSolve(A, B,free='t');
```

$$X := \begin{bmatrix} -3\,t_2 - 2\,t_4 \\ t_2 \\ 4\,t_2 + 3\,t_4 \\ t_4 \end{bmatrix}$$

```
> t[2]:=0;t[4]:=1;
```

$$t_2 := 0$$
$$t_4 := 1$$

```
> v1:=Transpose(X);
```

$$v1 := [-2,\ 0,\ 3,\ 1]$$

```
> t[2]:=1;t[4]:=0;
```

$$t_2 := 1$$
$$t_4 := 0$$

```
> v2:=Transpose(X);
```

$$v2 := [-3,\ 1,\ 4,\ 0]$$

Thus, the vectors $V_1 = (-2, 0, 3, 1)$ and $V_2 = (-3, 1, 4, 0)$ form a basis of L.

The second step is the orthogonalization of the basis $S = \{V_1, V_2\}$. We obtain the orthogonal basis $\{U_1, U_2\}$ of L where $U_1 = (-2, 0, 3, 1)$, $U_2 = (-3, 7, 1, -9)$.

The last step is the orthogonalization of the system $S = \{U_1, U_2, X\}$ (see Figure 1. There the vector V_l should be thought as X, the vector U_s corresponds to the resulting vector Y below). As a result we get the orthogonal system $\{U_1, U_2, Y\}$ where

$$Y = \left(\frac{3}{5}, \frac{13}{5}, -\frac{1}{5}, \frac{9}{5}\right) .$$

Consequently,

$$d = \| Y \| = \frac{1}{5}\sqrt{260} = \frac{2}{5}\sqrt{65} .$$

Computations in MAPLE

```
> with(LinearAlgebra):
```

```
> v1:=Vector[row]([-2,0,3,1]);
```

$$v1 := [-2,\ 0,\ 3,\ 1]$$

```
> v2:=Vector[row]([-3,1,4,0]);
```

$$v2 := [-3,\ 1,\ 4,\ 0]$$

```
> U:=GramSchmidt([v1,v2]);
```

$$U := \left[[-2,\ 0,\ 3,\ 1],\ \left[\frac{-3}{7},\ 1,\ \frac{1}{7},\ \frac{-9}{7}\right]\right]$$

```
> u1:=U[1];
```

$$u1 := [-2, 0, 3, 1]$$

```
> u2:=7*U[2]; # we scale the vector U[2] by a non-trivial factor
```

$$u2 := [-3, 7, 1, -9]$$

```
> X0:=Vector[row]([-1,5,1,-1]);
```

$$X0 := [-1, 5, 1, -1]$$

```
> U0:=GramSchmidt([u1,u2,X0]);
```

$$U0 := \left[[-2, 0, 3, 1], [-3, 7, 1, -9], \left[\frac{3}{5}, \frac{13}{5}, \frac{-1}{5}, \frac{9}{5}\right]\right]$$

```
> Y:=U0[3];
```

$$Y := \left[\frac{3}{5}, \frac{13}{5}, \frac{-1}{5}, \frac{9}{5}\right]$$

```
> distance:=Norm(Y,2);
```

$$distance := \frac{2}{5}\sqrt{65}$$

Remark. For this example we split the whole Maple worksheet into two parts according to the algorithm. It is more convenient to compute the distance using a single worksheet.

1.5. The matrix exponential function

Given a square matrix A with real or complex entries the matrix exponential function exp is defined as follows

$$\exp(A) = E + A + \frac{A^2}{2!} + \frac{A^3}{3!} + \ldots + \frac{A^n}{n!} + \ldots = \sum_{n=0}^{\infty} \frac{A^n}{n!} \,. \tag{1.21}$$

Here E stands for the identity matrix. The matrix series converges to the matrix $\exp(A)$ for any matrix A.

The exponential matrix plays a big role in solving linear systems of differential equations.

Example 1.8. Determine $\exp(A)$ if

$$A = \begin{pmatrix} \lambda_1 & 0 & 0 \\ 0 & \lambda_2 & 0 \\ 0 & 0 & \lambda_3 \end{pmatrix} .$$

Solution. Since

$$A^n = \begin{pmatrix} \lambda_1^n & 0 & 0 \\ 0 & \lambda_2^n & 0 \\ 0 & 0 & \lambda_3^n \end{pmatrix}$$

and we have the expansion series for any number $z \in \mathbb{C}$

$$e^z = \exp z = 1 + z + \frac{z^2}{2!} + \frac{z^3}{3!} + \ldots + \frac{z^n}{n!} + \ldots = \sum_{n=0}^{\infty} \frac{z^n}{n!} \,,$$

then it follows from the definition (1.21) that

$$\exp(A) = \begin{pmatrix} e^{\lambda_1} & 0 & 0 \\ 0 & e^{\lambda_2} & 0 \\ 0 & 0 & e^{\lambda_3} \end{pmatrix} .$$

Example 1.9. Determine $\exp(tA)$ (here $t \in \mathbb{R}$) if

$$A = \begin{pmatrix} 1 & 1 \\ -1 & 1 \end{pmatrix}.$$

Computations in MAPLE

```
> with(LinearAlgebra):
> A:=Matrix([[1,1],[-1,1]]);
```

$$A := \begin{bmatrix} 1 & 1 \\ -1 & 1 \end{bmatrix}$$

```
> with(linalg): exponential(A*t);
```

$$\begin{bmatrix} e^t \cos(t) & e^t \sin(t) \\ -e^t \sin(t) & e^t \cos(t) \end{bmatrix}$$

As a result we obtain the following matrix

$$\exp(tA) = \begin{pmatrix} e^t \cos t & e^t \sin t \\ -e^t \sin t & e^t \cos t \end{pmatrix}.$$

Example 1.10. Determine $\exp(tJ)$ if

$$J = \begin{pmatrix} \lambda & 1 & 0 & 0 \\ 0 & \lambda & 1 & 0 \\ 0 & 0 & \lambda & 1 \\ 0 & 0 & 0 & \lambda \end{pmatrix}.$$

Computations in MAPLE

```
> with(LinearAlgebra):
> J:=Matrix([[lambda,1,0,0],[0,lambda,1,0],[0,0,lambda,1],[0,0,0,lambda
> ]]);
```

$$J := \begin{bmatrix} \lambda & 1 & 0 & 0 \\ 0 & \lambda & 1 & 0 \\ 0 & 0 & \lambda & 1 \\ 0 & 0 & 0 & \lambda \end{bmatrix}$$

```
> with(linalg): exponential(J*t);
```

$$\begin{bmatrix} e^{(t\,\lambda)} & t\,e^{(t\,\lambda)} & \frac{1}{2}\,t^2\,e^{(t\,\lambda)} & \frac{1}{6}\,t^3\,e^{(t\,\lambda)} \\ 0 & e^{(t\,\lambda)} & t\,e^{(t\,\lambda)} & \frac{1}{2}\,t^2\,e^{(t\,\lambda)} \\ 0 & 0 & e^{(t\,\lambda)} & t\,e^{(t\,\lambda)} \\ 0 & 0 & 0 & e^{(t\,\lambda)} \end{bmatrix}$$

Thus,

$$\exp(tJ) = \begin{pmatrix} e^{t\lambda} & t\,e^{t\lambda} & \frac{t^2}{2!}\,e^{t\lambda} & \frac{t^3}{3!}\,e^{t\lambda} \\ 0 & e^{t\lambda} & t\,e^{t\lambda} & \frac{t^2}{2!}\,e^{t\lambda} \\ 0 & 0 & e^{t\lambda} & t\,e^{t\lambda} \\ 0 & 0 & 0 & e^{t\lambda} \end{pmatrix}.$$